

Capitolul 10. SQL Dinamic

Aproape orice dezvoltator de aplicații cu baze de date își dorește să poată lansa dintr-un bloc procedural o comandă SQL. Această facilitate a fost considerată un lux până prin versiunea 7 a Oracle când producătorii au furnizat un pachet special denumit DBMS_SQL. Deși un pas important, acest pachet este astăzi demodat, nu numai pentru că necesită etape destul de laborioase în lansarea unei comenzi SQL, ci și pentru că noile tipuri de date, introduse în versiunile 8x și 9x nu sunt recunoscute. Astfel încât, pentru prezentul capitol, ne-am propus să discutăm numai despre SQL Dinamic Nativ - NDS (*Native Dynamic SQL*), așa cum a apărut în Oracle 8i și se găsește în PL/SQL-ul “contemporan”.

Aria aplicațiilor NDS este departe de a fi limitată doar la execuția dintr-un bloc PL/SQL a unor comenzi SQL, iar în cele ce urmează vom parcurge și câteva exemple de apel dinamic al unor procedurii/funcții, interogări mono și multi-linie, recursivitate etc.

10.1. Comenzi DDL

În mod normal, PL/SQL-ul nu permite crearea/ștergerea de tabele sau alte obiecte din baza de date, definirea de restricții, acordarea de drepturi utilizatorilor și altor comenzi de tip DDL (*Data Definition Language*). Din Oracle 8i, comanda `EXECUTE IMMEDIATE` elimină această limitare. Blocul anonim din listing 10.1 prezintă două variante de creare a tabelii `CONCEDII` ce conține planificarea perioadelor de concedii de odihnă pentru fiecare angajat al firmei, attributele fiind marca, data inițială și data finală a concediului. Firește, concediul poate fi luat pe “bucăți”, cheia primară fiind combinația (`Marca`, `DataI`).

Listing 10.1. Două variante de creare a tabelii `CONCEDII`

```
DECLARE
    tabela VARCHAR2(30) ;
    atribut1 VARCHAR2(50) ;
    restrictii_atribut1 VARCHAR2(500) ;
    atribut2 VARCHAR2(50) ;
    restrictii_atribut2 VARCHAR2(500) ;
    atribut3 VARCHAR2(50) ;
    restrictii_atribut3 VARCHAR2(500) ;
    restrictie1 VARCHAR2(500) ;
    restrictie2 VARCHAR2(500) ;
    restrictie3 VARCHAR2(500) ;

BEGIN

    -- primul mod de creare a tabelii CONCEDII
    EXECUTE IMMEDIATE '
        CREATE TABLE concedii (
            marca INTEGER
```

```

        CONSTRAINT nn_concedii_marca NOT NULL
        CONSTRAINT fk_concedii_personal
            REFERENCES personal (marca)
, datai DATE
, dataf DATE
, CONSTRAINT ck_concedii1 CHECK (dataf >= datai)
, CONSTRAINT ck_concedii2 CHECK (dataf - datai < 90)
, CONSTRAINT pk_concedii PRIMARY KEY (marca,datai)
);

-- urmează a doua varianta

-- dar mai întâi, ștergem tabela creată mai sus
EXECUTE IMMEDIATE 'DROP TABLE concedii' ;

-- apoi o re-creăm, dar, de date aceasta, folosind variabile
tabela := 'concedii' ;
atribut1 := 'marca INTEGER' ;
restrictii_atribut1 := ' CONSTRAINT nn_concedii_marca NOT NULL
        CONSTRAINT fk_concedii_personal REFERENCES personal (marca) ' ;
atribut2 := 'datai DATE' ;
restrictii_atribut2 := ' ' ;
atribut3 := 'dataf DATE' ;
restrictii_atribut3 := ' ' ;
restrictie1 := ' CONSTRAINT ck_concedii1 CHECK (dataf >= datai) ' ;
restrictie2 := ' CONSTRAINT ck_concedii2 CHECK (dataf - datai < 90) ' ;
restrictie3 := ' CONSTRAINT pk_concedii PRIMARY KEY (marca,datai) ' ;

EXECUTE IMMEDIATE 'CREATE TABLE ' || tabela || ' ( ' ||
    atribut1 || restrictii_atribut1 ||
    ', ' || atribut2 || restrictii_atribut2 ||
    ', ' || atribut3 || restrictii_atribut3 ||
    ', ' || restrictie1 ||
    ', ' || restrictie2 ||
    ', ' || restrictie3 || ' ) ' ;

END ;
/

```

Comenzile SQL (CREATE TABLE, DROP TABLE, INSERT, UPDATE etc.) nu se termină cu punct-virgulă; în schimb, acest caracter marchează finalul comenzii EXECUTE IMMEDIATE. Prima variantă este cea în care, între apostrofuluri, se inserează întreaga comandă CREATE, ca și cum ar fi scrisă în SQL*Plus sau într-un script. A doua variantă folosește variabile, iar argumentul comenzii EXECUTE IMMEDIATE este o combinație de literali și variabile. Între cele două variante de EXECUTE IMMEDIATE cu CREATE TABLE a fost introdusă o comandă pentru ștergerea tabelului pentru a evita declanșarea unei erori datorate existenței tabelului ce s-ar dori a fi creat. Fără ștergere, a doua execuție a blocului s-ar fi soldat cu o eroare datorată încercării de a recrea o tabelă care există, după cum arată figura 10.1.

```

SQL> @f:\oracle_carte\cap10_sql_dinamic\listing10_01.sql
64 /

PL/SQL procedure successfully completed.

SQL> desc concedii
Name
-----
MARCA
DATAI
DATAF

SQL> @f:\oracle_carte\cap10_sql_dinamic\listing10_01.sql
64 /
DECLARE
*
ERROR at line 1:
ORA-00955: name is already used by an existing object
ORA-06512: at line 15

SQL>

```

Figura 10.1. Lansarea repetată a blocului PL/SQL ce conține comanda de creare a tabelui

De aceea, se poate redacta o procedură pentru generalizarea ștergerii oricărei tabeli, procedură care, pentru evitarea oricărei erori, verifică în prealabil existența tabeli ce urmează a fi eliminată din schemă. În secțiunea dedicată excepțiilor, se preia eventuala `NO_DATA_FOUND` și printr-un `NULL` elegant nu se execută nimic – vezi listing 10.2.

Listing 10.2. Procedura de ștergere dinamică a unei tabeli

```

CREATE OR REPLACE PROCEDURE p_sterge (
  tabela VARCHAR2)
IS
  v_unu NUMBER(1) := 0 ;
BEGIN
  SELECT 1 INTO v_unu FROM USER_TABLES
    WHERE table_name = UPPER(tabela) ;
  EXECUTE IMMEDIATE ' DROP TABLE ' || tabela ;
EXCEPTION
  WHEN NO_DATA_FOUND THEN
    -- tabela nu există, așa că e mai bine să n-o ștergem !
    NULL ;
END p_sterge ;

```

Apelul acestei proceduri se realizează de o manieră celei prezentate în blocul anonim din listing 10.3.

Listing 10.3. Apelul procedurii de ștergere

```

BEGIN
  p_sterge ('CONCEDII') ;

  EXECUTE IMMEDIATE '

```

```

CREATE TABLE concedii (
    marca INTEGER
        CONSTRAINT nn_concedii_marca NOT NULL
        CONSTRAINT fk_concedii_personal
            REFERENCES personal (marca)
    , datai DATE
    , dataf DATE
    , CONSTRAINT ck_concedii1 CHECK (dataf >= datai)
    , CONSTRAINT ck_concedii2 CHECK (dataf - datai < 90)
    , CONSTRAINT pk_concedii3 PRIMARY KEY (marca,datai)
);

END ;

```

Beneficiind de cunoștințele acumulate cu atâta efort, putem purcede la a construi un pachet, numit, cam pretențios, PAC_ADMINISTRARE, în care vom combina colecțiile de tip vectori asociativi, pe unul sau două nivele, cu SQL dinamic și alte ingrediente necesare. Specificațiile sale constituie subiectul listing-ului 10.4.

Listing 10.4. Specificațiile pachetului PAC_ADMINISTRARE

```

CREATE OR REPLACE PACKAGE pac_administrare AUTHID CURRENT_USER
AS

    TYPE t_atribute IS TABLE OF VARCHAR2 (200) INDEX BY PLS_INTEGER ;
    TYPE t_restrictii_atribute IS TABLE OF t_atribute INDEX BY PLS_INTEGER ;

    TYPE t_RefCursor IS REF CURSOR ;

    PROCEDURE p_creare_tabele (
        tabela VARCHAR2, v_atribute t_atribute, v_tip_atribute t_atribute,
        v_restrictii_atribute t_restrictii_atribute, v_restrictii_tabela t_atribute ) ;

    PROCEDURE p_sterge_tabela (tabela VARCHAR2) ;

    PROCEDURE p_dezactiveaza_restrictii (tabela VARCHAR2) ;

    PROCEDURE p_dezactiveaza_restrictii (v_restrictii t_atribute) ;

    PROCEDURE p_activeaza_restrictii (tabela VARCHAR2) ;

    PROCEDURE p_activeaza_restrictii (v_restrictii t_atribute) ;

END pac_administrare ;

```

Dacă `t_atribute` este un banal tip de tablou asociativ, componentele sale având o lungime de maxim 200 caractere, `t_restrictii_atribute` reprezintă un tip mai exotic – vector de vectori (nici o legătură cu jargonul manelisto-cartierist !); altfel spus, `t_restrictii` este un soi de matrice; primul indice va desemna linia atributului, iar coloana o restricție a atributului respectiv. Cele șase proceduri sunt destinate creării și ștergerii de tabele și activării/dezactivării de restricții. De fapt, sunt doar patru proceduri, deoarece s-a apelat la supraîncărcarea `P_DEZACTIVEAZA_RESTRICTII` și `P_ACTIVEAZA_RESTRICTII`. Listingul

10.5 conține impresionantele corpuri ale procedurilor. Cea mai “supraponderală” este procedura de creare dinamică a tabelelor. Aceasta deoarece procedura nu numai că articulează argumentele comenzii CREATE TABLE, ci și pentru că încearcă să dea un nume cât mai sugestiv fiecărei restricții.

Primul dintre cei cinci parametri de intrare, *tabela*, este de tip scalar și desemnează numele tabelului ce se dorește a fi creată dinamic. Ceilalți patru, *v_atribute*, *v_tip_atribute*, *v_restricții_atribute* și *v_restricții_tabela* sunt de tip colecție. Operațiunea de creare a tabelului este precedată de apelul procedurii de ștergere, pentru înlăturarea eventualei erori ce apare la încercarea de creare a unei tabeli deja existente.

Articularea opțiunilor comenzii CREATE TABLE se derulează în doi pași, unul care trece în revistă toate atributele, iar celălalt care parcurge toate restricțiile definite după ultimul atribut (de obicei, acestea sunt restricțiile la nivel de înregistrare). Primul pas se bazează pe “scanarea” tuturor componentelor din tabloul *v_atribute*. Pentru fiecare atribut, a fost introdusă o secvență iterativă pentru preluarea eventualelor restricții, prilej de folosire a tabeloului bidimensional *v_restricții_atribute*. Partea ce mai consistentă a acestei iterații ține de atribuirea automată a unui nume pentru restricția curentă. Fideli notației pseudo-ungurești, restricțiile de tip CHECK vor începe cu CK, de tip PRIMARY KEY cu PK, UNIQUE cu UN și REFERENCES cu FK.

Al doilea pas constă într-o altă secvență iterativă dedicată restricțiilor definite după ultimul atribut, restricții care se găsesc în tabloul *v_restricții_tabela*. Și aici un spațiu generos este dedicat “botezării” fiecărei restricții în funcție de tipul său. Întrucât pentru un atribut sau pentru o tabelă pot fi declarate mai multe clauze CHECK, respectiv CHECK/UNIQUE/REFERENCES, s-a recurs la variabilele *ultimul_check*, *ultimul_unique* și *ultimul_reference* pentru a evita repetările generatoare de erori.

Listing 10.5. Corpul pachetului PAC_ADMINISTRARE

```
CREATE OR REPLACE PACKAGE BODY pac_administrare AS
-----
PROCEDURE p_creare_tabele (
    tabela VARCHAR2, v_atribute t_atribute,
    v_tip_atribute t_atribute, v_restricții_atribute t_restricții_atribute,
    v_restricții_tabela t_atribute )
IS
    sir1 VARCHAR2 (4000) := '';
    sir2 VARCHAR2 (2000) := '';
    ultimul_check NUMBER(2) := 0 ;
    ultimul_unique NUMBER(2) := 0 ;
    ultimul_reference NUMBER(2) := 0 ;
    ultimul_atribut VARCHAR2(30) := NULL ;
    nume_restricție VARCHAR2(30) ;
BEGIN
    p_sterge_tabela (tabela) ;
    sir1 := 'CREATE TABLE ' || tabela || '(' ;

    -- se parcurg atributele
    FOR i IN 1..v_atribute.COUNT LOOP
        -- de la al doilea atribut introducem separatorul virgulă
```

```

IF i > 1 THEN
    sir1 := sir1 || ',' ;
END IF ;
sir1 := sir1 || v_atribute (i) || ' ' || v_tip_atribute (i) || ' ' ;

-- pentru fiecare atribut se trec în revistă restricțiile sale
FOR j IN 1..99 LOOP
    -- atenție la formatul EXISTS pentru vectorii bidimensionali !
    IF v_restrictii_atribute(i).EXISTS(j) THEN
        -- se caută un nume pentru restricție
        CASE
            WHEN v_restrictii_atribute(i)(j) LIKE '%PRIMARY KEY%' THEN
                nume_restricție := 'PK_' || tabela ;
            WHEN v_restrictii_atribute(i)(j) LIKE '%UNIQUE%' THEN
                nume_restricție := 'UN_' || tabela || '_' || v_atribute (i) ;
            WHEN v_restrictii_atribute(i)(j) LIKE '%NOT NULL%' THEN
                nume_restricție := 'NN_' || tabela || '_' || v_atribute (i) ;
            WHEN v_restrictii_atribute(i)(j) LIKE '%REFERENCES%' THEN
                nume_restricție := 'FK_' || tabela || '_' || v_atribute (i) ;
            WHEN v_restrictii_atribute(i)(j) LIKE '%CHECK%' THEN
                -- sunt posibile mai multe CHECK-uri pentru același atribut
                IF NVL(ultimul_atribut, '') = v_atribute (i) THEN
                    ultimul_check := ultimul_check + 1 ;
                ELSE
                    ultimul_atribut := v_atribute (i) ;
                    ultimul_check := 1 ;
                END IF ;
                nume_restricție := 'CK_' || tabela || '_' || v_atribute (i) || '_' ||
                    ultimul_check ;
            END CASE ;
        -- se introduce în comanda CREATE TABLE clauza CONSTRAINT
        sir1 := sir1 || ' CONSTRAINT ' || nume_restricție || ' ' ||
            v_restrictii_atribute(i)(j) || ' ' ;
    ELSE
        EXIT ;
    END IF ;
END LOOP ;
END LOOP ;

-- se trec în revistă eventualele restricții la nivel de înregistrare
ultimul_unique := 0 ;
ultimul_check := 0 ;
ultimul_reference := 0 ;
FOR i IN 1..99 LOOP
    IF v_restrictii_tabela.EXISTS(i) THEN
        -- mai întâi se caută un nume pentru restricție
        CASE
            WHEN v_restrictii_tabela(i) LIKE '%PRIMARY KEY%' THEN
                nume_restricție := 'PK_' || tabela ;
            WHEN v_restrictii_tabela(i) LIKE '%UNIQUE%' THEN
                -- sunt posibile mai multe UNIQUE-uri la nivel de tabelă
                ultimul_unique := ultimul_unique + 1 ;
                nume_restricție := 'UN_' || tabela || '_' || ultimul_unique ;
            WHEN v_restrictii_tabela(i) LIKE '%CHECK%' THEN
                -- sunt posibile mai multe CHECK-uri
                ultimul_check := ultimul_check + 1 ;
                nume_restricție := 'CK_' || tabela || '_' || ultimul_check ;
            WHEN v_restrictii_tabela(i) LIKE '%REFERENCES%' THEN
                -- sunt posibile mai multe REFERENCES-uri
                ultimul_reference := ultimul_reference + 1 ;

```

```

        nume_restricție := 'FK_' || tabela || '_' || ultimul_reference ;
    END CASE ;
    -- se introduce clauza CONSTRAINT pentru restricțiile la nivel de înregistrare
    sir2 := sir2 || ', CONSTRAINT ' || nume_restricție || ' ' || v_restricții_tabela(i) || ' ' ;
ELSE
    EXIT ;
END IF ;
END LOOP ;

sir2 := sir2 || ' ' ;

-- și acum, crearea !!!
EXECUTE IMMEDIATE ' ' || sir1 || sir2 ;

```

```
END p_creare_tabele ;
```

```

-----
PROCEDURE p_sterge_tabela (tabela VARCHAR2)
IS
    v_unu NUMBER(1) := 0 ;
BEGIN
    SELECT 1 INTO v_unu FROM USER_TABLES
        WHERE table_name = UPPER(tabela) ;
    EXECUTE IMMEDIATE ' DROP TABLE ' || tabela ;
EXCEPTION
    WHEN NO_DATA_FOUND THEN
        -- tabela nu există, așa că e mai bine să n-o ștergem !
        NULL ;
END p_sterge_tabela ;

```

```

-----
PROCEDURE p_dezactiveaza_restricții (tabela VARCHAR2)
IS
    v_restricții t_atribute ;
BEGIN
    SELECT constraint_name BULK COLLECT INTO v_restricții
        FROM user_constraints WHERE table_name = UPPER(RTRIM(tabela)) ;
    FOR i IN 1..v_restricții.COUNT LOOP
        EXECUTE IMMEDIATE 'ALTER TABLE ' || tabela || ' DISABLE CONSTRAINT '
            || v_restricții(i) ;
    END LOOP ;
END p_dezactiveaza_restricții ;

```

```

-----
PROCEDURE p_dezactiveaza_restricții (v_restricții t_atribute )
IS
    tabela VARCHAR2 (50) ;
BEGIN
    FOR i IN 1..v_restricții.COUNT LOOP
        SELECT table_name INTO tabela FROM user_constraints
            WHERE constraint_name = UPPER(RTRIM(v_restricții(i))) ;
        EXECUTE IMMEDIATE 'ALTER TABLE ' || tabela || ' DISABLE CONSTRAINT '
            || v_restricții(i) ;
    END LOOP ;
END p_dezactiveaza_restricții ;

```

```

-----
PROCEDURE p_activeaza_restricții (tabela VARCHAR2)
IS

```

```

    v_restricii t_attribute ;
BEGIN
    SELECT constraint_name BULK COLLECT INTO v_restricii
        FROM user_constraints WHERE table_name = UPPER(RTRIM(tabela)) ;
    FOR i IN 1..v_restricii.COUNT LOOP
        EXECUTE IMMEDIATE 'ALTER TABLE ' || tabela || ' ENABLE CONSTRAINT '
            || v_restricii(i) ;
    END LOOP ;
END p_activeaza_restricii ;

-----
PROCEDURE p_activeaza_restricii (v_restricii t_attribute )
IS
    tabela VARCHAR2 (50) ;
BEGIN
    FOR i IN 1..v_restricii.COUNT LOOP
        SELECT table_name INTO tabela FROM user_constraints
            WHERE constraint_name = UPPER(RTRIM(v_restricii(i))) ;
        EXECUTE IMMEDIATE 'ALTER TABLE ' || tabela || ' ENABLE CONSTRAINT '
            || v_restricii(i) ;
    END LOOP ;
END p_activeaza_restricii ;

END pac_administrare ;

```

Celelalte proceduri ale pachetului sunt ceva mai simple. Și pentru P_ACTIVEAZA_RESTRICȚII și pentru P_DEZACTIVEAZA_RESTRICȚII s-a recurs s-a supraîncărcare. Artificiul supraîncărcării este util pentru determinarea restricțiilor de activat/dezactivat. Astfel, dacă parametrul de intrare al procedurii este simplu (șir de caractere), atunci este vorba de o tabelă și se activează/dezactivează toate restricțiile declarate pentru tabela respectivă. Când parametrul este de tip colecție (vector asociativ), procedura “pricepe” că este vorba de o listă de restricții ce trebuie activate sau dezactivate.

Ceea ce rămâne în sarcina procedurilor este ca, atunci când parametrul de intrare este scalar (tabela), să extragă din USER_CONSTRAINTS toate restricțiile declarate pentru tabela respectivă, iar atunci când parametrul este de tip colecție, să extragă, pentru fiecare restricție în parte, tabela pentru care a fost definită restricția respectivă. Astfel, se cunosc toți parametrii comenzilor ALTER TABLE *tabelă* ENABLE CONSTRAINT *restricție* și ALTER TABLE *tabelă* DISABLE CONSTRAINT *restricție* care se vor lansa prin EXECUTE IMMEDIATE.

Blocul anonim din listing 10.6 ilustrează apelul procedurii de creare “dinamică” a tabelii CONCEDII. Variabilele tablou locale, v_atrib, v_tip_atrib, v_restr_atrib și v_restr_tab sunt de tipurile declarate în specificațiile pachetului PAC_ADMINISTRARE.

Listing 10.6. Crearea unei tabeli folosind pachetul PAC_ADMINISTRARE

```

DECLARE
    v_atrib pac_administrare.t_attribute ;
    v_tip_atrib pac_administrare.t_attribute ;
    v_restr_atrib pac_administrare.t_restricii_attribute ;
    v_restr_tab pac_administrare.t_attribute ;

```



```
BEGIN
```

```
-- atributele și restricțiile lor
v_atrib(1) := 'marca' ;
v_tip_atrib(1) := 'INTEGER' ;
v_restr_atrib(1)(1) := 'NOT NULL' ;
v_restr_atrib(1)(2) := 'REFERENCES personal (marca)' ;

v_atrib(2) := 'datai' ;
v_tip_atrib(2) := 'DATE' ;
v_restr_atrib(2)(1) := 'NOT NULL' ;
v_restr_atrib(2)(2) := 'CHECK (EXTRACT (YEAR FROM datai) >= 2003) ' ;

v_atrib(3) := 'dataf' ;
v_tip_atrib(3) := 'DATE' ;
v_restr_atrib(3)(1) := 'NOT NULL' ;

-- restricțiile la nivel de tabela
v_restr_tab (1) := 'CHECK (dataf >= datai)' ;
v_restr_tab (2) := 'CHECK (dataf - datai < 90)' ;
v_restr_tab (3) := 'PRIMARY KEY (marca, datai)' ;

-- și acum, marele apel !!!
pac_administrare.p_creare_tabele ('CONCEDII', v_atrib, v_tip_atrib, v_restr_atrib,
v_restr_tab) ;
```

```
END ;
```

Firește, specificarea corectă a tipului fiecărui atribut și restricțiilor la nivel de câmp sau înregistrare cade în sarcina celui care redactează blocul de apel al procedurii. Lansarea în execuție a acestui bloc, ca și vizualizarea ulterioară din SQL*Plus a celor trei atribute ale tabelului CONCEDII, este ilustrată în figura 10.2.

```
SQL> @f:\oracle_carte\cap10_sql_dinamic\listing10_06.sql
33 /
```

```
PL/SQL procedure successfully completed.
```

```
SQL> desc concedii
```

```
  Name
```

```
-----
```

```
  MARCA
```

```
  DATAI
```

```
  DATAF
```

```
SQL>
```

Figura 10.2. Apelul procedurii de creare a tabelului și vizualizarea atributelor

Privitor la folosirea procedurilor de activare și dezactivare de restricții din schema curentă, avem pregătit un nou bloc, la fel de anonim – vezi listing 10.7. Blocul combină apelul procedurilor din pachet cu secvențe iterative pentru afișarea stării restricțiilor, tocmai pentru a pune în evidență rezultatele procedurilor. După o primă afișare a stării restricțiilor, se apelează P_DEZACTIVEAZA_RESTRICTII care dezactivează toate restricțiile la nivel de

atribut și înregistrare ale tabelii CONCEDII. Se afișează starea după apel, iar apoi se lansează procedura pentru activarea tuturor restricțiilor aceleiași tabeli. Secvența iterativă pentru afișare este urmată de o dezactivare selectivă – numai a restricțiilor PK_CONCEDII și FK_CONCEDII_MARCA. Cu ultimele puteri, blocul mai afișează odată starea restricțiilor.

Listing 10.7. Activări și dezactivări de restricții folosind SQL dinamic

```

DECLARE
    v_restr pac_administrare.t_atribute ;
BEGIN
    DBMS_OUTPUT.PUT_LINE ('INAINTE de dezactivare, restrictiile tabelii CONCEDII:');
    FOR rec_1 IN (SELECT constraint_name, status FROM user_constraints
                  WHERE table_name = 'CONCEDII') LOOP
        DBMS_OUTPUT.PUT_LINE ('Restrictie '|| rec_1.constraint_name ||
                              ', stare:'||rec_1.status);
    END LOOP ;

    -- se dezactivează toate restricțiile tabelii CONCEDII
    pac_administrare.p_dezactiveaza_restrictii('CONCEDII');
    DBMS_OUTPUT.PUT_LINE (' ');
    DBMS_OUTPUT.PUT_LINE ('DUPA dezactivare, restrictiile tabelii CONCEDII:');
    FOR rec_1 IN (SELECT constraint_name, status FROM user_constraints
                  WHERE table_name = 'CONCEDII') LOOP
        DBMS_OUTPUT.PUT_LINE ('Restrictie '|| rec_1.constraint_name
                              || ', stare:'||rec_1.status);
    END LOOP ;

    -- se RE-activează toate restricțiile tabelii CONCEDII
    pac_administrare.p_activeaza_restrictii('CONCEDII');
    DBMS_OUTPUT.PUT_LINE (' ');
    DBMS_OUTPUT.PUT_LINE ('Dupa RE-activare, restrictiile tabelii CONCEDII:');
    FOR rec_1 IN (SELECT constraint_name, status FROM user_constraints
                  WHERE table_name = 'CONCEDII') LOOP
        DBMS_OUTPUT.PUT_LINE ('Restrictie '|| rec_1.constraint_name
                              || ', stare:'||rec_1.status);
    END LOOP ;

    v_restr (1) := 'PK_CONCEDII' ;
    v_restr (2) := 'FK_CONCEDII_MARCA' ;
    pac_administrare.p_dezactiveaza_restrictii(v_restr) ;

    DBMS_OUTPUT.PUT_LINE (' ');
    DBMS_OUTPUT.PUT_LINE
        ('Dupa dezactivarea celor doua, restrictiile tabelii CONCEDII:');
    FOR rec_1 IN (SELECT constraint_name, status FROM user_constraints
                  WHERE table_name = 'CONCEDII') LOOP
        DBMS_OUTPUT.PUT_LINE ('Restrictie '|| rec_1.constraint_name
                              || ', stare:'||rec_1.status);
    END LOOP ;
END ;

```

Rezultatul SQL*Plus al acestor laborioase operațiuni este afișat în figura 10.3.

```

Oracle SQL*Plus
File Edit Search Options Help

SQL> @f:\oracle_carte\cap10_sql_dinamic\listing10_07.SQL
44 /
INAINTE de dezactivare, restrictiile tablei CONCEDEII:
Restrictie NN_CONCEDII_MARCA, stare:ENABLED
Restrictie NN_CONCEDII_DATAI, stare:ENABLED
Restrictie NN_CONCEDII_DATAF, stare:ENABLED
Restrictie CK_CONCEDII_DATAI_1, stare:ENABLED
Restrictie CK_CONCEDII_1, stare:ENABLED
Restrictie CK_CONCEDII_2, stare:ENABLED
Restrictie PK_CONCEDII, stare:ENABLED
Restrictie FK_CONCEDII_MARCA, stare:ENABLED
DUPA dezactivare, restrictiile tablei CONCEDEII:
Restrictie NN_CONCEDII_MARCA, stare:DISABLED
Restrictie NN_CONCEDII_DATAI, stare:DISABLED
Restrictie NN_CONCEDII_DATAF, stare:DISABLED
Restrictie CK_CONCEDII_DATAI_1, stare:DISABLED
Restrictie CK_CONCEDII_1, stare:DISABLED
Restrictie CK_CONCEDII_2, stare:DISABLED
Restrictie PK_CONCEDII, stare:DISABLED
Restrictie FK_CONCEDII_MARCA, stare:DISABLED
Dupa RE-activare, restrictiile tablei CONCEDEII:
Restrictie NN_CONCEDII_MARCA, stare:ENABLED
Restrictie NN_CONCEDII_DATAI, stare:ENABLED
Restrictie NN_CONCEDII_DATAF, stare:ENABLED
Restrictie CK_CONCEDII_DATAI_1, stare:ENABLED
Restrictie CK_CONCEDII_1, stare:ENABLED
Restrictie CK_CONCEDII_2, stare:ENABLED
Restrictie PK_CONCEDII, stare:ENABLED
Restrictie FK_CONCEDII_MARCA, stare:ENABLED
Dupa dezactivarea celor doua, restrictiile tablei CONCEDEII:
Restrictie NN_CONCEDII_MARCA, stare:ENABLED
Restrictie NN_CONCEDII_DATAI, stare:ENABLED
Restrictie NN_CONCEDII_DATAF, stare:ENABLED
Restrictie CK_CONCEDII_DATAI_1, stare:ENABLED
Restrictie CK_CONCEDII_1, stare:ENABLED
Restrictie CK_CONCEDII_2, stare:ENABLED
Restrictie PK_CONCEDII, stare:DISABLED
Restrictie FK_CONCEDII_MARCA, stare:DISABLED

PL/SQL procedure successfully completed.

SQL> |

```

Figura 10.3. Apeluri ale procedurilor de activare/dezactivare a restricțiilor

10.2. Apelul “dinamic” al unei proceduri/funcții

Cel de-al doilea ingredient din SQL dinamic nativ pe care îl discutăm succint în continuare este legat de crearea și lansarea dintr-o procedură a unui bloc PL/SQL, anonim și mai puțin anonim. Câteva explicații despre obiectivul urmărit nu strică. În aplicațiile client/server tradiționale, pe două straturi, sesiunea de lucru este descrisă la logarea clientului cu numele utilizator și parola corespunzătoare, iar obiectele instanțiate în memorie pot fi accesate în limita drepturilor acordate. În aplicațiile web, însă, dialogul utilizatorului cu baza de date este mijlocit de serverul de aplicații care preia pe cont propriu cererile clienților și negociază de unul singur cu serverul de bază de date. Aceasta

înseamnă că sesiunea server de aplicații – server de bază de date este partajată de mai mulți clienți.

10.2.1. Crearea și folosirea dinamică a unei pachet

Procedura următoare, P_PACHET_USER (listing 10.8), care primește drept parametru de intrare numele utilizatorului, crează un pachet ce conține variabila publică `data_ultimei_modificari`. Interesant este că numele pachetului conține și numele utilizatorului: `PACHET_utilizator`, ceea ce reprezintă o premieră pentru cartea de față. Practic, procedura verifică existența pachetului, folosind tabela virtuală din dicționar `USER_OBJECTS`. Blocul inclus preia excepția `NO_DATA_FOUND` care semnalizează inexistența pachetului în schema curentă și, printr-un `EXECUTE IMMEDIATE` fulgerător, crează dinamic pachetul al cărui nume nu este mai puțin... dinamic. La revenirea în blocul principal se apelează (cum altfel, decât dinamic ?) pachetul, modificându-se valoarea variabilei publice cu data și ora curentă.

Listing 10.8. Crearea și apelul dinamic al unui pachet

```
CREATE OR REPLACE PROCEDURE p_pachet_user (utilizator VARCHAR2)
AUTHID CURRENT_USER
AS
    v_unu NUMBER(1);
BEGIN
    BEGIN
        SELECT 1 INTO v_unu
        FROM user_objects
        WHERE object_type='PACKAGE' AND object_name = 'PACHET_'||utilizator ;
    EXCEPTION
        WHEN NO_DATA_FOUND THEN
            EXECUTE IMMEDIATE 'CREATE OR REPLACE PACKAGE pachet_' || utilizator
                || ' AUTHID CURRENT_USER AS data_ultimei_modificari DATE ;
                END pachet_' || utilizator ||';' ;
    END ;
    EXECUTE IMMEDIATE 'BEGIN pachet_' || utilizator ||
        '.data_ultimei_modificari := SYSDATE; END ;';
END p_pachet_user ;
```

Prin urmare, un bloc PL/SQL poate fi inclus într-o comandă `EXECUTE IMMEDIATE` "liniarizându-l", corectitudinea sintactică a blocului (atenție la terminatorul de comandă, punct-virgulă !) fiind decisivă pentru succesul apelului. Figura 10.4 suprinde în SQL*Plus starea dinaintea apelului procedurii, lansarea de două ori a procedurii (cu argumentele 'MARIN1' și 'MARIN2'), apelul dinamic al pachetului pentru modificarea valorii variabilei publice `data_ultimei_modificari` și, în final, afișarea pachetelor nou create din dicționar (`USER_OBJECTS`).

Se cuvine de remarcat că data creării și a ultimei modificări a celor două pachete nu indică tocmai adevărul...

Din păcate, variabilele declarate în bloc nu pot fi accesate prin comanda `EXECUTE IMMEDIATE`. Astfel, blocul din listing 10.9 conține două variabile, `v1` și `v2`, prima fiind inițializată chiar din secțiunea declarațiilor.

```
SQL> SELECT object_name, created, last_ddl_time, status
  2 FROM user_objects WHERE object_type = 'PACKAGE'
  3 AND object_name LIKE 'PACHET_MARIN%'
  4 /

no rows selected

SQL> EXECUTE p_pachet_user('MARIN1')

PL/SQL procedure successfully completed.

SQL> EXECUTE p_pachet_user('MARIN2')

PL/SQL procedure successfully completed.

SQL> BEGIN
  2 pachet_marin1.data_ultimei_modificari := SYSDATE + 1;
  3 END;
  4 /

PL/SQL procedure successfully completed.

SQL> SELECT object_name, created, last_ddl_time, status
  2 FROM user_objects WHERE object_type = 'PACKAGE'
  3 AND object_name LIKE 'PACHET_MARIN%'
  4 /
```

OBJECT_NAME	CREATED	LAST_DDL_	STATUS
PACHET_MARIN1	01-APR-03	01-APR-03	VALID
PACHET_MARIN2	01-APR-03	01-APR-03	VALID

```
SQL> |
```

Figura 10.4. Ilustrarea, în SQL Plus, a folosirii procedurii `P_PACHET_USER` și apelului dinamic al unui pachet

Blocul PL/SQL executat dinamic încearcă să atribuie o valoare variabilei `v2`.

Listing 10.9. Domeniul de vizibilitate pentru variabile din bloc PL/SQL dinamic

```
DECLARE
  v1 VARCHAR2(100) := 'Valoare 1';
  v2 VARCHAR2(100);
BEGIN
  EXECUTE IMMEDIATE 'BEGIN v2:= ""||"Valoare 2"||""; END ;';
  DBMS_OUTPUT.PUT_LINE(v2);
END ;
```

Lansarea în execuție prilejuiește primirea unui mesaj de eroare, după cum se observă în figura 10.5. Mesajul este limpede. Variabila `v2` nu este recunoscută de comanda `EXECUTE IMMEDIATE`; prin urmare toate variabilele incluse în blocul inclus în comandă au regim privat (local).

```

SQL> DECLARE
  2   v1 VARCHAR2(100) := 'Valoare 1' ;
  3   v2 VARCHAR2(100) ;
  4 BEGIN
  5   EXECUTE IMMEDIATE ' BEGIN v2:= ''||'Valoare 2'||'' ; END ; ' ;
  6   DBMS_OUTPUT.PUT_LINE(v2) ;
  7 END ;
  8 /
DECLARE
*
ERROR at line 1:
ORA-06550: line 1, column 8:
PLS-00201: identifier 'v2' must be declared
ORA-06550: line 1, column 8:
PL/SQL: Statement ignored
ORA-06512: at line 5

```

Figura 10.5. Vizibilitatea unei variabile incluse în EXECUTE IMMEDIATE

Firește, acest neajuns poate fi depășit dacă se recurge la variabile publice (din pachet). Execuția listingului 10.10 crează, în prima etapă, specificațiile unui pachet, PAC_STERGE ce conține două variabile, inevitabil publice, v1 și v2. Blocul anonim care urmează pasează, mai întâi, variabilei v2 valoarea din v1, printr-o banală comandă de atribuire, apoi modifică aceeași valoare printr-un bloc anonim inclus într-o comandă EXECUTE IMMEDIATE.

Listing 10.10. Apelarea dintr-un bloc dinamic a variabilei din pachet

```

CREATE OR REPLACE PACKAGE pac_sterge AS
  v1 VARCHAR2(100) := 'Valoare 1';
  v2 VARCHAR2(100);
END pac_sterge ;
/
BEGIN
  pac_sterge.v2:= pac_sterge.v1 ;
  EXECUTE IMMEDIATE ' BEGIN pac_sterge.v2 := ""||'Valoare 2'||"" ; END ; ' ;
  DBMS_OUTPUT.PUT_LINE(pac_sterge.v2) ;
END ;

```

Rezultatul (figura 10.6) este încurajator.

```

SQL> CREATE OR REPLACE PACKAGE pac_sterge AS
  2   v1 VARCHAR2(100) := 'Valoare 1' ;
  3   v2 VARCHAR2(100) ;
  4 END pac_sterge ;
  5 /

Package created.

SQL>
SQL> BEGIN
  2   pac_sterge.v2:= pac_sterge.v1 ;
  3   EXECUTE IMMEDIATE ' BEGIN pac_sterge.v2:= ''||'Valoare 2'||'' ; END ; ' ;
  4   DBMS_OUTPUT.PUT_LINE(pac_sterge.v2) ;
  5 END ;
  6 /
Valoare 2

PL/SQL procedure successfully completed.

SQL>

```

Figura 10.6. Apelul, din blocul dinamic, a unei variabile publice

10.2.2. Preluarea valorilor unei înregistrări și generarea dinamică a comenzii INSERT

În cele ce urmează ne propunem un obiectiv ceva mai ambițios. În aplicațiile client-server sau web, deseori inserarea, modificarea sau ștergerea de înregistrări din tabele nu se realizează direct prin comenzi `INSERT/UPDATE/DELETE` lansate de pe client către serverul de baze de date, ci prin apelul unei proceduri de inserare/modificare/ștergere, procedură căreia i se transmit sub formă de parametri, valorile înregistrării de adăugat sau cheile primare/condițiile liniilor de modificat sau șters.

Să discutăm cazul tablei `PERSONAL`, oprindu-ne doar la operațiunea de adăugare a unei înregistrări. Interfața aplicației preia valorile atributelor și le pasează sub forma unei variabile compozite (de tip înregistrare – `ROWTYPE`). Atenție, însă, dacă valoarea pasată pentru un atribut este `NULL`, vrem ca în tabelă atributul respectiv să primească valoarea implicită (clauza `DEFAULT` din comanda `CREATE TABLE` sau `ALTER TABLE`). Or, dacă `INSERT` ar avea formatul `INSERT INTO personal VALUES (valoare1, valoare2, ...)`, oricare valoare `NULL` ar putea intra în conflict cu o eventuală restricție `NOT NULL` declarată pentru atributul respectiv. De aceea, procedura pe care o vom redacta în continuare va testa, pe rând, valoarea fiecărui atribut. Numai dacă aceasta este nenulă, atributul respectiv va fi inclus în comanda de inserare dinamică. Practic, în final, se va executa o comandă de genul `INSERT INTO personal (atribut_nenul1, atribut_nenul2,...) VALUES (valoare_nenulă1, valoare_nenulă2,)`.

Listingul 10.11 conține atât specificațiile, cât și corpul pachetului `PAC_INSERT`. Variabilele publice sunt:

- `v_enull` care va conține `TRUE` dacă valoarea atributului curent este nulă și `FALSE` în caz contrar;
- `v_sir` ce conține valoarea nenulă a unui atribut sub formă de șir ce urmează a fi inclus în clauza `VALUES`;
- `v_personal` – o copie a tuturor valorilor atributelor pasate la apelul procedurii `P_INSERT`;

Soluția propusă se bazează, în bună măsură, pe supraîncărcarea funcției `F_VAL` care, în funcție de tipul atributului primit, va returna șirul de caractere (inclusiv, apostrofuri, funcții `TO_DATE`) necesar clauzei `VALUES`. În fine, procedura ce generează comanda `INSERT` se numește `P_INSERT` și prezintă un parametru de intrare de tip înregistrare (`personal%ROWTYPE`).

Listing 10.11. Pachetul pentru generarea dinamică a comenzii INSERT

```
CREATE OR REPLACE PACKAGE pac_insert AS
  v_enull BOOLEAN := FALSE ;
  v_sir VARCHAR2(100);
```

```

    v_personal personal%ROWTYPE ;
FUNCTION f_val (atribut_ VARCHAR2) RETURN VARCHAR2 ;
FUNCTION f_val (atribut_ NUMBER) RETURN VARCHAR2 ;
FUNCTION f_val (atribut_ DATE) RETURN VARCHAR2 ;
PROCEDURE p_insert (rec_personal personal%ROWTYPE) ;
END pac_insert ;
/

CREATE OR REPLACE PACKAGE BODY pac_insert AS

-- prelucrarea valorilor de tip ȘIR DE CARACTERE
FUNCTION f_val (atribut_ VARCHAR2) RETURN VARCHAR2 IS
BEGIN
    RETURN "" || atribut_ || "";
END f_val ;

-- prelucrarea valorilor de tip NUMERIC
FUNCTION f_val (atribut_ NUMBER) RETURN VARCHAR2 IS
BEGIN
    RETURN '' || atribut_ || '' ;
END f_val ;

-- prelucrarea valorilor de tip DATA CALENDARISTICA
FUNCTION f_val (atribut_ DATE) RETURN VARCHAR2 IS
BEGIN
    RETURN ' TO_DATE (" || TO_CHAR(atribut_,'DD/MM/YYYY') || " , "DD/MM/YYYY") ' ;
END f_val ;

/* procedura de preluare a valorilor unei înregistrări din PERSONAL și generare
dinamică a comenzii INSERT */
PROCEDURE p_insert (rec_personal personal%ROWTYPE) IS
    sir_atribute VARCHAR2 (2000) ; -- INSERT INTO personal (atribut_i, atribut_j...)
    sir_valori VARCHAR2 (2000) ; -- VALUES (valoarea_i, valoare_j, ...)
BEGIN
    v_personal := rec_personal ; -- copierea valorilor în variabila din pachet

    -- se parcurg toate atributele tabelului și se testează care au valori NULLE
    FOR rec_atribute IN (SELECT column_name, column_id, data_type
        FROM USER_TAB_COLUMNS WHERE TABLE_NAME ='PERSONAL'
        ORDER BY column_id) LOOP
        IF rec_atribute.column_id = 1 THEN -- primul atribut
            sir_atribute := ' INSERT INTO PERSONAL ( ' ;
            sir_valori := ' VALUES ( ' ;
        END IF ;

        -- se testează dacă valoarea atributului curent este NULL
        -- artificial consta în folosirea unui bloc dinamic care atribuie valoarea
        -- TRUE sau FALSE variabilei publice V_ENULL
        EXECUTE IMMEDIATE
            ' BEGIN IF pac_insert.v_personal.' || rec_atribute.column_name ||
                ' IS NULL THEN pac_insert.v_enull := TRUE ;
                ELSE pac_insert.v_enull := FALSE ; END IF; END; ' ;
        IF v_enull = TRUE THEN
            -- se omite din lista atributelor și cea a valorilor
            NULL ;
        ELSE
            -- valoarea atributului curent nu este NULL, deci trebuie modificate și
            -- lista atributelor și cea a valorilor
            IF rec_atribute.column_id > 1 THEN
                sir_atribute := sir_atribute || ', ' ;
            
```



```

        sir_valori := sir_valori || ',';
    END IF;
    -- se adaugă numele în lista atributelor
    sir_atribute := sir_atribute || rec_atribute.column_name;

    /* un alt artificiu, cel mai interesant: valoarea atributului curent se pasează
    unei funcții F_VAL din pachet; aceasta va returna variabilei publice V_SIR
    șirul de caractere ce trebuie inclus în clauza VALUES din comanda INSERT;
    întrucât valoarea atributului poate fi numerică, șir de caractere sau
    dată calendaristică, ne bazăm pe supraîncărcarea funcției F_VAL */
    EXECUTE IMMEDIATE ' BEGIN pac_insert.v_sir :=
        pac_insert.f_val ( pac_insert.v_personal.||
        rec_atribute.column_name || ' '); END; ';
    -- se adaugă valoarea V_SIR în lista valorilor clauzei VALUES
    sir_valori := sir_valori || pac_insert.v_sir;
    END IF;
END LOOP;
sir_atribute := sir_atribute || ')';
sir_valori := sir_valori || ')';

-- pentru impresie, se afișează cele două părți ale comenzii INSERT
DBMS_OUTPUT.PUT_LINE(sir_atribute);
DBMS_OUTPUT.PUT_LINE(sir_valori);

-- în fine, ECCE INSERT-ul !
EXECUTE IMMEDIATE sir_atribute || ' ' || sir_valori;
END p_insert;
END pac_insert;
/

```

La SQL dinamic s-a recurs în trei puncte ale procedurii. Mai întâi, se testează dacă valoarea atributului curent (`pac_insert.v_personal.|| rec_atribute.column_name`) este NULLă, rezultatul fiind stocat în variabila publică `v_enull`. Al doilea moment este cel al apelului dinamic al funcției `F_VAL`, pasându-i-se valoarea atributului curent. Este și momentul în care intră în acțiune supraîncărcarea, deoarece tipul valorii transmise determină instanța funcției ce va intra în execuție. În fine, ultima comandă `EXECUTE IMMEDIATE` este cea care lansează comanda `INSERT`.

Pentru testarea funcției, apelăm la blocul anonim din listing 10.12. Acesta inițializează valorile atributelor unei înregistrări din `PERSONAL` și i le pasează procedurii din pachetul `PAC_INSERT`.

Listing 10.12. Apelul procedurii de inserare dinamică

```

DECLARE
    linie_personal personal%ROWTYPE;
BEGIN
    -- initializarea valorii atributelor
    linie_personal.marca := 1011;
    linie_personal.numepren := 'Angajat 1011';
    linie_personal.compart := NULL;
    linie_personal.datasv := TO_DATE ('12/05/1997', 'DD/MM/YYYY');
    linie_personal.salorar := NULL;
    linie_personal.salorarco := NULL;
    linie_personal.colaborator := NULL;

```

```

-- apelul procedurii
pac_insert.p_insert (linie_personal) ;
COMMIT ;
END ;
/

```

Valorile atributelor compart, salorar, salorarco și colaborator sunt nule, așa ca acestea ar trebui excluse în formatul comenzii INSERT. Apelul blocului de mai sus din SQL*Plus este prezentat în figura 10.7.

```

SQL> DECLARE
2   linie_personal personal%ROWTYPE ;
3   BEGIN
4   -- initializarea valorii atributelor
5   linie_personal.marca := 1011 ;
6   linie_personal.numepren := 'Angajat 1011' ;
7   linie_personal.compart := NULL ;
8   linie_personal.datasv := TO_DATE ('12/05/1997', 'DD/MM/YYYY') ;
9   linie_personal.salarar := NULL ;
10  linie_personal.salararco := NULL ;
11  linie_personal.colaborator := NULL ;
12
13  -- apelul procedurii
14  pac_insert.p_insert (linie_personal) ;
15  commit ;
16  END ;
17  /
INSERT INTO PERSONAL ( MARCA, NUMEPREN, DATASV )
VALUES ( 1011 , 'Angajat 1011', TO_DATE ('12/05/1997', 'DD/MM/YYYY') )

PL/SQL procedure successfully completed.

SQL> |

```

Figura 10.7. Apelul procedurii de inserare dinamică

Cele două linii afișate la execuție constituie părțile comenzii INSERT. Așa cum ne-am propus, în listă sunt incluse numai atributele ale căror valori sunt nenule (marca, numepren și datasv).

10.3. Variabile legate și comenzi DML dinamice

Procedura P_PACHET_USER se poate redacta și de maniera prezentată în listing 10.13. Elementul de noutate îl constituie folosirea unei variabile legate. Astfel, conținutul atribuit variabilei din pachet nu este specificat direct, ci prin variabila legată referită prin :1, clauza USING fiind cea care face efectiv legătura :1 - SYSDATE.

Listing 10.13. O primă variabilă legată

```

CREATE OR REPLACE PROCEDURE p_pachet_user (utilizator VARCHAR2)
AUTHID CURRENT_USER AS
v_unu NUMBER(1);
BEGIN

```

```

BEGIN
    ....
END ;
EXECUTE IMMEDIATE 'BEGIN pachet_|| utilizator || '.data_ultimei_modificari := :1 ;
    END ;' USING SYSDATE ;
END p_pachet_user ;

```

Firește, nimic nu se schimbă în apelul procedurii, după cum reiese și din figura 10.8. Apelul procedurii crează pachetul PACHET_MARIN3 și inițializează variabila din acesta folosind variabila legată.

```

SQL> EXECUTE p_pachet_user('MARIN3')

PL/SQL procedure successfully completed.

SQL> SELECT object_name, created, last_ddl_time, status
2  FROM user_objects WHERE object_type = 'PACKAGE'
3  AND object_name LIKE 'PACHET_MARIN%'
4  /

```

OBJECT_NAME	CREATED	LAST_DDL_	STATUS
PACHET_MARIN1	01-APR-03	01-APR-03	VALID
PACHET_MARIN2	01-APR-03	01-APR-03	VALID
PACHET_MARIN3	01-APR-03	01-APR-03	VALID

```

SQL> BEGIN
2  DBMS_OUTPUT.PUT_LINE(pachet_marin3.data_ultimei_modificari);
3  END;
4  /
01-APR-03

PL/SQL procedure successfully completed.

SQL> |

```

Figura 10.8. Afișarea valorii setate prin folosirea variabilei legate

Să luăm un alt exemplu în care variabile legate se folosesc pentru o comandă `INSERT` “dinamică”. Blocul anonim din listing 10.14 prezintă, pentru comparație, o variantă de inserare “obișnuită”, în care constantele de tip șir de caractere sunt încadrate de două apostrofuri (nu confundați cu ghilimelele !) și o alta, nu neapărat neobișnuită, în care se folosesc patru variabile legate.

Listing 10.14. Inserare dinamică, fără și cu variabile legate

```

DECLARE
    v_marca personal.marca%TYPE := 1006;
    v_numepren personal.numepren%TYPE := 'Angajat 6';
    v_compart personal.compart%TYPE := 'CONTA';
    v_datasv personal.datasv%TYPE := TO_DATE ('23/11/1989', 'DD/MM/YYYY');
BEGIN
    -- varianta 1 - FĂRĂ variabile legate
    EXECUTE IMMEDIATE 'INSERT INTO personal (marca, numepren, compart, datasv)
        VALUES (1005, "Angajat 5", "IT", TO_DATE("12/05/1992", "DD/MM/YYYY"))';

    -- varianta 2 - CU variabile legate
    EXECUTE IMMEDIATE 'INSERT INTO personal (marca, numepren, compart, datasv)

```

```
VALUES (:1, :2, :3, :4 ) ' USING v_marca, v_numepren, v_compart, v_datasv ;
COMMIT ;
END ;
```

Analog inserării, se pot folosi variabile legate în EXECUTE IMMEDIATE pentru modificări și ștergeri de linii. Una dintre cele mai importante restricții ale variabilelor legate ține de faptul că acestea nu pot desemna nume de obiecte ale bazei. Modificăm ușor forma blocului din listingul precedent și încercăm ca în comanda INSERT să specificăm numele tabelului PERSONAL printr-o variabilă legată – vezi listing 10.15.

Listing 10.15. Variabilă legată (incorect) pentru specificarea numelui tabelului

```
DECLARE
  v_marca personal.marca%TYPE := 1099;
  v_numepren personal.numepren%TYPE := 'Angajat 99' ;
  v_compart personal.compart%TYPE := 'CONTA' ;
  v_datasv personal.datasv%TYPE := TO_DATE ('20/09/1997', 'DD/MM/YYYY') ;
  v_tabela VARCHAR2 (30) := 'PERSONAL' ;
BEGIN
  -- tentativa nr.3
  EXECUTE IMMEDIATE 'INSERT INTO :1 (marca, numepren, compart, datasv)
    VALUES (:2, :3, :4, :5 ) ' USING v_tabela, v_marca, v_numepren, v_compart, v_datasv ;
END ;
```

Consecințele sunt dezaastroase – figura 10.9. După cum anticipam, PL/SQL se opune energic desemnării numelui oricărui obiect al bazei (în cazul nostru, tabelă) printr-o variabilă legată.

```
SQL> DECLARE
2  v_marca personal.marca%TYPE := 1099;
3  v_numepren personal.numepren%TYPE := 'Angajat 99' ;
4  v_compart personal.compart%TYPE := 'CONTA' ;
5  v_datasv personal.datasv%TYPE := TO_DATE ('20/09/1997', 'DD/MM/YYYY') ;
6  v_tabela VARCHAR2 (30) := 'PERSONAL' ;
7  BEGIN
8  -- tentativa nr.3
9  EXECUTE IMMEDIATE 'INSERT INTO :1 (marca, numepren, compart, datasv)
10 VALUES (:2, :3, :4, :5 ) ' USING v_tabela, v_marca, v_numepren, v_compart, v_datasv ;
11 END ;
12 /
DECLARE
*
ERROR at line 1:
ORA-00903: invalid table name
ORA-06512: at line 9
```

Figura 10.9. Folosire eronată a unei variabile legate

Acest gen de situație reclamă utilizarea variabilelor pentru desemnarea tabelului de maniera prezentată în paragraful 10.1 – vezi și listing 10.16.

Listing 10.16. Sintaxa corectă de specificare, prin variabilă, a numelui tabelului

```
DECLARE
  v_marca personal.marca%TYPE := 1099;
  v_numepren personal.numepren%TYPE := 'Angajat 99' ;
  v_compart personal.compart%TYPE := 'CONTA' ;
  v_datasv personal.datasv%TYPE := TO_DATE ('20/09/1997', 'DD/MM/YYYY') ;
```

```

v_tabela VARCHAR2 (30) := 'PERSONAL' ;
BEGIN
  -- tentativa nr.3
  EXECUTE IMMEDIATE 'INSERT INTO ' || v_tabela ||
    ' (marca, numepren, compart, datasv) VALUES (:1, :2, :3, :4) '
    USING v_marca, v_numepren, v_compart, v_datasv ;
END ;

```

10.4. Interogări dinamice mono-linie

Un exemplu tipic de funcție dinamică cu interogare al cărei rezultat este mono-linie vizează furnizarea numărului de înregistrări dintr-o tabelă specificată¹ - vezi listing 10.15. De data aceasta, clauza folosită este `INTO`, prin care se precizează în ce variabile se va stoca rezultatul interogării.

Listing 10.17. Funcție ce furnizează numărul de linii dintr-o tabelă specificată

```

CREATE OR REPLACE FUNCTION f_nr_linii (
  tabela IN VARCHAR2,
  conditie IN VARCHAR2 ) RETURN INTEGER
IS
  v_rezultat INTEGER;
BEGIN
  EXECUTE IMMEDIATE 'SELECT COUNT(*) FROM ' || tabela || ' WHERE ' ||
    NVL (conditie, '1=1') INTO v_rezultat ;
  RETURN v_rezultat ;
END;

```

Listing 10.18 conține două blocuri anonime pentru apelul funcției. La primul apel se transmite doar numele tabelii – `TRANSE_SV`. Al doilea apel valorifică și avantajul specificării unei condiții pe care trebuie să o îndeplinească liniile tabelii pentru a fi luate în calcul la numărare. Și în aceste situații trebuie mărită atenția la modul în care sunt introduși literalii, respectiv modul de folosire a apostrofului.

Listing 10.18. Două blocuri de apel a funcției `F_NR_LINII`

```

BEGIN
  DBMS_OUTPUT.PUT_LINE('In tabela ' || 'TRANSE_SV' ||
    ' sunt ' || f_nr_linii('TRANSE_SV', NULL)
    || ' linii ');
END ;
/

BEGIN
  DBMS_OUTPUT.PUT_LINE('In tabela ' || 'PERSONAL' ||
    ' sunt ' || f_nr_linii('PERSONAL', 'compart = "IT"')
    || ' angajat in compartimentul IT ');
END ;
/

```

¹ Vezi și [Fuerstein???] [Fotache??? – sql dinamic]

Figura 10.10 prezintă rezultatele lansării celor două blocuri anonime PL/SQL din SQL*Plus.

```

Oracle SQL*Plus
File Edit Search Options Help

SQL> BEGIN
2  DBMS_OUTPUT.PUT_LINE('In tabela ' || 'TRANSE_SV' ||
3  ' sunt ' || f_nr_linii('TRANSE_SV', NULL)
4  || ' linii ');
5  END ;
6  /
In tabela TRANSE_SV sunt 6 linii

PL/SQL procedure successfully completed.

SQL>
SQL> BEGIN
2  DBMS_OUTPUT.PUT_LINE('In tabela ' || 'PERSONAL' ||
3  ' sunt ' || f_nr_linii('PERSONAL', 'compart = ''IT''')
4  || ' angajat in compartimentul IT ');
5  END ;
6  /
In tabela PERSONAL sunt 5 angajat in compartimentul IT

PL/SQL procedure successfully completed.

SQL>
SQL>

```

Figura 10.10. Două apeluri a funcției F_NR_LINII

Pentru divertisment, ne propunem să obținem tabela ce are cel mai mare număr de înregistrări din schema curentă, scop pentru care vom redacta o frază SELECT ce apelează funcția de mai sus:

```

SELECT *
FROM      (SELECT table_name AS tabela,
                f_nr_linii(table_name, NULL) AS nr_inreg
            FROM USER_TABLES ) cite_inreg
WHERE nr_inreg =
      (SELECT MAX(nr_inreg)
       FROM
        (SELECT f_nr_linii(table_name, NULL)
         AS nr_inreg
         FROM USER_TABLES )
       )

```

Revenim la blocul anonim PL/SQL din listing 10.10. Pentru a modifica valoarea variabilei v2 din pachetul PAC_STERGE, putem folosi și un mic artificiu, ca în listing 10.19. Artificiul este interogarea ce “preia” datele din tabela sistem DUAL. Expresia din clauza SELECT va fi pasată variabilei specificate prin opțiunea INTO a comenzii EXECUTE IMMEDIATE.

Listing 10.19. O altă modalitate de modificare a variabilei din pachet

```

BEGIN
  pac_sterge.v2:= pac_sterge.v1 ;
  EXECUTE IMMEDIATE 'SELECT '''||'Valoare 2'||''' FROM dual' INTO pac_sterge.v2 ;

```

```
DBMS_OUTPUT.PUT_LINE(pac_sterge.v2);
END ;
```

10.5. 10.5. Interogări multi-linie

Comanda `EXECUTE IMMEDIATE` a fost cea care a rezolvat, până în prezent, toate problemele în care a fost necesar SQL dinamic. A rămas o singură speță, cea a interogărilor ale căror rezultat numără mai multe linii, pentru care for recurge la o formă modificată a comenzii `OPEN` aplicată variabilelor cursor.

10.5.1. Arhivarea tabeli PONTAJE pe ani calendaristici

Pentru cele ce urmează ne propunem crearea unei proceduri destinate arhivării anuale a pontajelor. Deoarece volumul tabeli `PONTAJE` poate căpăta, în timp, proporții impresionante și, pe de altă parte, doar rareori se lucrează cu pontaje de pe ani anteriori, mulți responsabili de aplicații preferă arhivarea datelor unui an într-o tabelă de genul *tabelă_an*, și apoi descongestionarea tabeli "master". Astfel, la începutul anului 2003, administratorul BD salvează toate pontajele care au data de întocmire cuprinsă între 1 ianuarie și 31 decembrie 2002 într-o tabelă numită `PONTAJE_2002`, după care se șterg liniile respective din tabela `PONTAJE`.

Procedura `ARHIVARE_PONTAJE` din pachetul `PAC_ARHIVARE` (listing 10.20) pare a rezolva problema la cel mai nepretențios mod cu putință. I se pasează anul (*nnnn*) pentru care se dorește arhivarea și, în caz că nu există (verificarea se face apelând la un bloc inclus), se creează tabela `PONTAJE_nnnn`. După verificare/creare, se inserează toate liniile din `PONTAJE` corespunzătoare anului arhivat, apoi se descongestionează tabela sursă de liniile proaspăt arhivate.

Listing 10.20. Arhivarea simplă a pontajelor dintr-un an

```
CREATE OR REPLACE PACKAGE pac_arhivare AUTHID CURRENT_USER AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE) ;
END pac_arhivare ;
/
CREATE OR REPLACE PACKAGE BODY pac_arhivare AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE)
IS
    v_unu NUMBER(1) := 0 ;
BEGIN
    -- se verifică dacă există PONTAJE_nnnn
    BEGIN
        SELECT 1 INTO v_unu FROM user_tables WHERE table_name = 'PONTAJE_' || an_ ;
    EXCEPTION
        WHEN NO_DATA_FOUND THEN -- nu există, așa că se creează tabela
            EXECUTE IMMEDIATE 'CREATE TABLE pontaje_' || an_ ||
                ' AS SELECT * FROM pontaje WHERE 1=2' ;

        -- se declară cheia primară a noii tabele
        EXECUTE IMMEDIATE 'ALTER TABLE pontaje_' || an_ ||
            ' ADD CONSTRAINT pk_pontaje_' || an_ || ' PRIMARY KEY (marca, data)' ;
    END ;
```

```

-- se inserează în PONTAJE_nnnn liniile din PONTAJE
EXECUTE IMMEDIATE 'INSERT INTO pontaje_'|| an_||
' SELECT * FROM pontaje WHERE TO_NUMBER(TO_CHAR(data, "YYYY")) = :1'
      USING an_ ;
-- se șterg din PONTAJE liniile arhivate
DELETE FROM pontaje WHERE TO_NUMBER(TO_CHAR(data, 'YYYY')) = an_ ;
COMMIT ;
END arhivare_pontaje ;

END pac_arhivare ;
/

```

Această soluție ridică unele probleme. Dacă, spre exemplu, după ce s-a operat arhivarea, se descoperă că unelele dintre înregistrările din PONTAJE ec au apucat să fie arhivate erau eronate (valorile atributelor *OreLucrate* și *OreCo* trebuiau inversate), acestea vor fi re-inserate în PONTAJE cu valorile corecte. La apelul următor al procedurii de arhivare, s-ar declanșa însă o eroare datorată violării cheii primare din PONTAJE_2002. Așa încât modificăm logica procedurii, astfel:

- se verifică existența tabelii PONTAJE_nnnn;
- se actualizează eventualele înregistrări din PONTAJE_nnnn, dar care prezintă corespondent PONTAJE, modificându-se atributele *OreLucrate*, *OreCo*, *OreNoapte*, *OreAbsNem*;
- se inserează în PONTAJE_nnnn liniile din PONTAJE inexistente în tabela arhivă;
- se șterg liniile din PONTAJE inserate sau actualizate în PONTAJE_nnnn.

La așa ambiții, trebuie modificată procedura de arhivare. Dintre variante, alegem una care să folosească SQL dinamic pentru o interogare multi-linie – vezi listing 10.21. Variabila cursor este denumită *v_cursor*. Ideea de bază a noii versiuni este ca, pentru actualizarea liniilor din PONTAJE_nnnn, să se încarce în variabila cursor toate înregistrările care sunt prezente în PONTAJE_nnnn de la o arhivare mai veche, dar care apar și în PONTAJE (se consideră că odată ce re-apar în PONTAJE, înregistrările trebuie actualizate în arhivă).

Practic, sâmburele de SQL-ul dinamic este legat de deschiderea variabilei cursor (*OPEN v_cursor FOR ...*) prin care în fraza *SELECT* se specifică denumirea tabelii din care se extrag datele la momentul execuției procedurii. În rest, încărcarea unei linii din cursor și secvența iterativă de parcurgere a tuturor liniilor din variabila cursor nu prezintă diferențe semnificative față de un cursor “clasic”.

Listing 10.21. Interogarea dinamică multi-linie

```

CREATE OR REPLACE PACKAGE pac_arhivare AUTHID CURRENT_USER AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE) ;
END pac_arhivare ;
/

CREATE OR REPLACE PACKAGE BODY pac_arhivare AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE)
IS

```



```

TYPE t_refcursor IS REF CURSOR ; -- declararea tipului variabilă-cursor
v_cursor t_refcursor ;          -- aceasta este variabila cursor
rec_cursor pontaje%ROWTYPE ;     -- variabila înregistrare pentru stocarea unei linii
                                -- din variabila-cursor

v_unu NUMBER(1) := 0 ;
TYPE t_pontaje IS RECORD (orelucrate pontaje.orelucrate%TYPE, oreco pontaje.oreco%TYPE,
orenoapte pontaje.orenoapte%TYPE, oreabsnem pontaje.oreabsnem%TYPE ) ;
v_pontaje t_pontaje ;
BEGIN
  BEGIN
    SELECT 1 INTO v_unu FROM user_tables WHERE table_name = 'PONTAJE_' || an_ ;
  EXCEPTION
    WHEN NO_DATA_FOUND THEN -- nu există, așa că se creează tabela
      EXECUTE IMMEDIATE 'CREATE TABLE pontaje_' || an_ ||
        ' AS SELECT * FROM pontaje WHERE 1=2 ' ;
      -- se declară cheia primară a noii tabelă
      EXECUTE IMMEDIATE 'ALTER TABLE pontaje_' || an_ ||
        ' ADD CONSTRAINT pk_pontaje_' || an_ || ' PRIMARY KEY (marca, data) ' ;
  END ;

  -- se extrag eventualele linii din PONTAJE_nnnn care există în PONTAJE pentru a fi actualizate
  OPEN v_cursor FOR 'SELECT marca, data FROM pontaje_' || an_ ||
    ' WHERE (marca, data) IN (SELECT marca, data FROM pontaje) ' ;
  LOOP
    FETCH v_cursor INTO rec_cursor ;
    EXIT WHEN v_cursor%NOTFOUND ;
    EXECUTE IMMEDIATE 'SELECT orelucrate, oreco, orenoapte, oreabsnem
      FROM pontaje WHERE marca = :1 AND data = :2 '
      INTO v_pontaje USING rec_cursor.marca, rec_cursor.data ;

    EXECUTE IMMEDIATE 'UPDATE pontaje_' || an_ || ' SET orelucrate = :1 , oreco = :2,
      orenoapte = :3, oreabsnem = :4 WHERE marca= :5 AND data = :6 '
      USING v_pontaje.orelucrate, v_pontaje.oreco, v_pontaje.orenoapte,
        v_pontaje.oreabsnem, rec_cursor.marca, rec_cursor.data ;
  END LOOP ;
  CLOSE v_cursor ;

  -- se insereaza în PONTAJE_nnnn liniile din PONTAJE ce nu există în prima tabelă
  EXECUTE IMMEDIATE 'INSERT INTO pontaje_' || an_ ||
    ' SELECT * FROM pontaje WHERE TO_NUMBER(TO_CHAR(data, "YYYY")) = :1
      AND (marca, data) NOT IN (SELECT marca, data FROM pontaje_' || an_ || ' ) ' USING an_ ;

  -- se șterg din PONTAJE liniile arhivate
  DELETE FROM pontaje WHERE TO_NUMBER(TO_CHAR(data, 'YYYY')) = an_ ;

  COMMIT ;
END arhivare_pontaje ;

END pac_arhivare ;
/

```

Merită remarcată, tot ca element de relativă noutate, și folosirea simultană, în prima comandă `EXECUTE IMMEDIATE` din buclă, a clauzelor `INTO` și `USING`. Prima servește la indicarea variabilei (variabilelor) în care se va salva rezultatul interogării dinamice, iar `USING` specifică variabilele legate, ale căror valori vor substitui parametrii `:1` și `:2`. Cea de-a doua comandă `EXECUTE IMMEDIATE` din secvența iterativă folosește variabile legate deopotrivă pentru indicarea

valorilor de preluat (în clauza `SET` a comenzii `UPDATE`) și predicatului de selecție (clauza `WHERE`).

10.5.2. Fuzionarea tabelelor arhivă

Arhivarea pontajelor pe ani calendaristici este benefică pentru descongestionarea tabeli `PONTAJE`, ținând seama că, de obicei, se editează pontajele de pe luna curentă și, eventual, luna anterioară. Dacă însă baza de date este operațională din 1999, iar conducerea firmei dorește o analiză a pontajelor pentru un interval calendaristic cuprins între 1 februarie 1999 și 20 februarie 2001, este necesară fuzionarea înregistrărilor a trei tabele: `PONTAJE_1999` (pentru înregistrările de după 1 februarie), `PONTAJE_2000` și `PONTAJE_2001` (numai înregistrările cuprinse între 1 ianuarie și 20 februarie). De aceea, este nevoie de o procedură specială care să fuzioneze oricâte tabele, în funcție de intervalul ales.

Procedura `FUZIUNE_PONTAJE` este inclusă în pachetul `PAC_ARHIVARE` și prezintă doi parametri de intrare, `data_initiala` și `data_finala`. Fuziunea se realizează într-o tabelă tampon numită `PONTAJE_TEMP`. Blocul inclus la începutul secțiunii executabile testează existența acestei tabele tampon și, dacă este cazul, o crează. Secvența iterativă se execută pentru fiecare an calendaristic din interval. La fiecare an se folosesc două variabile. O variabilă indică data inițială pentru anul respectiv – `data_in_an_crt` (data inițială an curent) – care este 1 ianuarie pentru toți anii din interval, cu excepția primului, și o altă variabilă – `data_sf_an_crt` – care este 31 decembrie, cu excepția ultimului an.

Listing 10.22. Fuziunea tabelelor `PONTAJE_nnnn`

```
CREATE OR REPLACE PACKAGE pac_arhivare AUTHID CURRENT_USER AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE) ;
PROCEDURE fuziune_pontaje (data_initiala DATE, data_finala DATE) ;
END pac_arhivare ;
/

CREATE OR REPLACE PACKAGE BODY pac_arhivare AS

PROCEDURE fuziune_pontaje (data_initiala DATE, data_finala DATE)
IS
    an_inceput NUMBER (4) ; -- primul an din intervalul de fuziune
    an_sfinit NUMBER (4) ; -- ultimul an din intervalul de fuziune
    data_in_an_crt DATE ; -- data de inceput din anul curent
    data_sf_an_crt DATE ; -- data de final din anul curent
    TYPE rc_tabela IS REF CURSOR ;
    c_tabela rc_tabela ;
    tab_ VARCHAR2(30) ;
BEGIN
    -- folosim o tabelă - PONTAJE_TEMP, pe care o creăm (dacă e nevoie)
    BEGIN
        SELECT table_name INTO tab_ FROM user_tables
        WHERE table_name = 'PONTAJE_TEMP' ;
        EXECUTE IMMEDIATE 'DELETE FROM ' || tab_ ;
    EXCEPTION
    WHEN NO_DATA_FOUND THEN
        EXECUTE IMMEDIATE 'CREATE TABLE pontaje_temp AS SELECT * FROM pontaje
```

```

        WHERE 1=2' ;
END ;
-- după acest bloc, PONTAJE_TEMP există și n-are înregistrări

-- extragerea anilor inițial și final ai intervalului calendaristic
an_inceput := TO_NUMBER(TO_CHAR(data_iniciala, 'YYYY')) ;
an_sfirsit := TO_NUMBER(TO_CHAR(data_finala, 'YYYY')) ;

FOR i IN an_inceput .. an_sfirsit LOOP          -- bucla executată pentru fiecare an din interval
    tab_ := 'PONTAJE_' || i ;
    -- cursorul C_TABELA conține cel mult o linie, dacă există tabela PONTAJE_nnnn,
    -- unde nnnn = i
    OPEN c_tabela FOR 'SELECT table_name FROM user_tables
        WHERE table_name = :1 ' USING tab_ ;
    FETCH c_tabela INTO tab_ ;

    IF c_tabela%ROWCOUNT > 0 THEN
        -- există tabelă-archivă pentru acest an
        -- pentru anul curent, datele inițiale și finale implicite sunt 1 ian. și 31 dec.
        data_in_an_crt := TO_DATE('01/01/'||i, 'DD/MM/YYYY') ;
        data_sf_an_crt := TO_DATE('31/12/'||i, 'DD/MM/YYYY') ;

        -- dacă e primul an din interval, data inițială ia valoarea DATA_IN
        IF i = an_inceput THEN
            data_in_an_crt := data_iniciala ;
        END IF ;

        -- dacă e ultimul an din interval, data finală ia valoarea DATA_SF
        IF i = an_sfirsit THEN          -- este ultimul an din interval
            data_sf_an_crt := data_finala ;
        END IF ;

        -- inserarea liniilor din PONTAJE_nnnn în PONTAJE_TEMP
        EXECUTE IMMEDIATE 'INSERT INTO pontaje_temp SELECT * FROM pontaje_' ||
            i || ' WHERE data BETWEEN :1 AND :2 ' USING data_in_an_crt, data_sf_an_crt ;
    END IF ;
END LOOP ;
COMMIT ;
END ;

-----
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE)
IS
    ... vezi listing anterior
END arhivare_pontaje ;

END pac_arhivare ;

```

10.5.3. SQL dinamic “înmănunchiat”

Revenim la această traducere exotică (pe care am lansat-o cu mult tam-tam în capitolul 8) pentru a introduce câteva opțiuni care ar putea ameliora sensibil performanțele blocurilor în care este folosit SQL-ul dinamic. Reamintim celor care au trecut în mare viteză prin sau pe lângă finalul capitolului 8 că înmănunchierea în PL/SQL presupune încărcarea sau parcurgerea unei variabile de tip colecție – vector asociativ, tabel încapsulat, vector cu mărime variabilă - dintr-o singură

mişcare, astfel încât transferul repetat al controlului, în cadrul unei secvențe repetitive, între motorul SQL și motorul PL/SQL este eliminat.

Trei comenzi constituie baza SQL-ului dinamic înmănușiat: EXECUTE IMMEDIATE, FETCH și FORALL. Listing-ul 10.23 ilustrează două dintre acestea. Pentru a le pune în valoare, schimbăm logica procedurii de arhivare. Astfel, valorile cheilor primare ale eventualelor înregistrări din PONTAJE_nnnn sunt salvate în două variabile de tip colecție – tablourile asociative v_marca și v_data – folosindu-se opțiunea BULK COLLECT INTO. Ulterior, în loc de a parcurge printr-o secvență iterativă toate componentele celor două variabile-masiv, actualizarea liniilor din PONTAJE_nnnn este realizată printr-o singură comandă, întrebuițându-se comanda FORALL... EXECUTE IMMEDIATE.

Listing 10.23. Folosirea clauzelor BULK COLLECT și FORALL

```
CREATE OR REPLACE PACKAGE pac_arhivare AUTHID CURRENT_USER AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE) ;
END pac_arhivare ;
/

CREATE OR REPLACE PACKAGE BODY pac_arhivare AS
PROCEDURE arhivare_pontaje (an_salarii.an%TYPE)
IS
    v_unu NUMBER(1) := 0 ;
    TYPE t_marca IS TABLE OF pontaje.marca%TYPE INDEX BY PLS_INTEGER ;
    v_marca t_marca ;
    TYPE t_data IS TABLE OF pontaje.data%TYPE INDEX BY PLS_INTEGER ;
    v_data t_data ;
BEGIN
    BEGIN
        ... identic listingului 10.21
    END ;

    -- se salvează în doi vectori cheile primare ale înregistrărilor ce urmează a fi actualizate
    EXECUTE IMMEDIATE 'SELECT marca, data FROM pontaje_'||an_||' WHERE (marca, data) IN
        (SELECT marca, data FROM pontaje) ' BULK COLLECT INTO v_marca, v_data ;

    FORALL i IN 1..v_marca.COUNT EXECUTE IMMEDIATE
        'UPDATE pontaje_'||an_||' p1
        SET (orelucrate, oreco, orenoapte, oreabsnem) =
            (SELECT orelucrate, oreco, orenoapte, oreabsnem FROM pontaje p2
             WHERE p2.marca = p1.marca AND p1.data = p2.data)
        WHERE marca = :1 AND data = :2 ' USING v_marca(i), v_data(i) ;

    -- se insereaza în PONTAJE_nnnn liniile din PONTAJE ce nu există în prima tabelă
    ... și apoi se șterg din PONTAJE liniile arhivate
    ... vezi listing 10.21

    COMMIT ;
END arhivare_pontaje ;

END pac_arhivare ;
```